#Region " Imports "

Imports System.Data
Imports BlazeApps.Library.Template

#End Region

Partial Class Usercontrols_Template_TemplateList
    Inherits UserControlBase

#Region " Enums, Memebers and Properties "

    Private DataPageLocal As New PagedDataSource

    Public Property sortCriteria() As String
        Get
            Return CStr(viewstate("sortCriteria"))
        End Get

        Set(ByVal Value As String)
            ViewState("sortCriteria") = Value
        End Set
    End Property

    Public Property sortDir() As String
        Get
            Return CStr(viewstate("sortDir"))
        End Get

        Set(ByVal Value As String)
            ViewState("sortDir") = Value
        End Set
    End Property

    Public Property DataLoaded() As Boolean
        Get
            Return GetBooleanFromVS("DataLoaded")
        End Get
        Set(ByVal value As Boolean)
            ViewState.Add("DataLoaded", value)
        End Set
    End Property

#End Region

#Region " Page Events "

    Protected Sub Page_Init(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Init
        Me.MessageLabel = Me.uxMessagesLabel
    End Sub

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        SetPageTitle("Template List")

        If Me.Page.IsPostBack = False Then
            LoadPage()
        End If

        ConfigBaseNavCtrls()

        If Me.Page.IsPostBack = False Then
            'Set Default Sort Order and Direction
            Me.sortCriteria = "templatename"
            Me.sortDir = "asc"

            BindDataItems()
        End If

        Me.uxTemplatePanel.Visible = ShowControls(SecurityMask.Template, True)

    End Sub

#End Region

#Region " Save Data "



#End Region

#Region " Load Data "

    Private Sub LoadPage()
        LoadPageSize(Me.uxPageSizeDropDownList)
    End Sub

    Public Overloads Overrides Sub BindDataItems()
        Dim TemplateList As New Template()
        Dim DT As New DataTable
        Dim DV As New DataView

        DT = TemplateList.GetTemplateList()

        Me.RecordCount = DT.Rows.Count

        DV.Table = DT

        DV.Sort = Me.sortCriteria & " " & Me.sortDir

        Me.DataPage.DataSource = DV
        Me.DataPage.AllowPaging = True
        Me.DataPage.PageSize = CType(Me.uxPageSizeDropDownList.SelectedValue.ToString, Integer)
        Me.DataPage.CurrentPageIndex = Me.PageNumber

        CreatePagingLinks()

        DataLoaded = True

        With Me.uxResultsGridView
            .DataSource = Me.DataPage
            .DataBind()
        End With

        DataLoaded = False

    End Sub

#End Region

#Region " Form Controls Events "

    Protected Sub uxNewTemplateButton_Click(ByVal sender As Object, ByVal e As System.EventArgs)
        Response.Redirect(String.Format("{0}.aspx?pid=tpl_template", AppPage))
    End Sub

    Protected Sub uxResultsGridView_RowCreated(ByVal sender As Object, ByVal e As System.Web.UI.WebControls.GridViewRowEventArgs) Handles uxResultsGridView.RowCreated

        If DataLoaded = True Then
            Dim LB As LinkButton = CType(e.Row.FindControl("uxDeleteTemplateLinkButton"), LinkButton)

            If e.Row.RowType = DataControlRowType.DataRow Then

                If e.Row.DataItem("templateusage").ToString = "0" Then
                    LB.Enabled = True
                Else
                    LB.Enabled = False
                End If
            End If
        End If

    End Sub

    Protected Sub DeleteTemplate(ByVal sender As Object, ByVal e As CommandEventArgs)
        TemplateDetail.DeleteTemplateDetail(ConvertToGuid(e.CommandArgument))
        BindDataItems()
    End Sub

    Protected Sub uxResultsGridView_Sorting(ByVal sender As Object, ByVal e As System.Web.UI.WebControls.GridViewSortEventArgs) Handles uxResultsGridView.Sorting
        'Get the Sort by Value from the Sortexpression field on the column
        If Me.sortCriteria = e.SortExpression Then
            If Me.sortDir = "desc" Then
                Me.sortDir = "asc"
            Else
                Me.sortDir = "desc"
            End If
        End If

        ' Assign the column clicked to the sortCriteria property
        Me.sortCriteria = e.SortExpression

        'Calls the function to rebuild the datagrid view, with the new sort order
        BindDataItems()
    End Sub

#End Region

#Region " Helpers and Methods "

    Private Sub ConfigBaseNavCtrls()

        Me.PagingTopPanel = Me.uxPagingTopPanel
        Me.PagingBottomPanel = Me.uxPagingBottomPanel

        Me.RepeaterTop = Me.uxPagesTopRepeater
        Me.RepeaterBottom = Me.uxPagesBottomRepeater

        Me.NextTopLinkButton = Me.uxTopNextLinkButton
        Me.PreviousTopLinkButton = Me.uxTopPrevLinkButton
        Me.NextBottomLinkButton = Me.uxBottomNextLinkButton
        Me.PreviousBottomLinkButton = Me.uxBottomPrevLinkButton

        Me.DataPage = DataPageLocal
        Me.PageSize = CType(Me.uxPageSizeDropDownList.SelectedValue.ToString, Integer)
        Me.PagesToDisplay = 10

        Me.ItemsPerPageDropDown = Me.uxPageSizeDropDownList
        Me.ItemsPerPageLabel = Me.uxItemsPerPageLabel
        Me.PagerInfoLabel = Me.uxPagerInfoLabel

        AddHandler Me.RepeaterTop.ItemCommand, AddressOf PagingRepeater_ItemCommand
        AddHandler Me.RepeaterBottom.ItemCommand, AddressOf PagingRepeater_ItemCommand

    End Sub

#End Region

End Class
